```
> restart;
> read "qFPS.mpl"; with(qFPS):
```

Equation (64), Proposition 9

```
> RE:=4*p^(4*k+10)*(p^4-1)^3*(p^(4*k)-p^(4*n))*(a^2*c^2*p^(4*n+4*k)-1)*d(k)+2*p^(4*n+2*k+5)*(p^4-1)^3*(p^2+1)*(p^(4*k+4)-1)*(c*p^(2*k+1)-1)*(a*p^(2*k+1)-1)*(a*c*p^(4*k+2)-1)*d(k+1)+p^(4*n)*(p^4-1)^3*(p^(4*k+4)-1)*(p^(4*k+8)-1)*(c*p^(2*k+1)-1)*(c*p^(2*k+3)-1)*(a*p^(2*k+1)-1)*(a*p^(2*k+3)-1)*d(k+2)=0;
```

$$RE := 4\,p^{(4k+10)}\,(p^4-1)^3\,(p^{(4k)}-p^{(4n)})\,(a^2c^2p^{(4n+4k)}-1)\,d(k)+2\,p^{(4n+2k+5)}\,(p^4-1)^3\,(p^2+1)\,(p^{(4k+4)}-1)\,(c\,p^{(2k+1)}-1)\,(a\,p^{(2k+1)}-1)\,(a\,c\,p^{(4k+2)}-1)\,d(k+1)+p^{(4n)}\,(p^4-1)^3\,(p^{(4k+4)}-1)\,(p^{(4k+8)}-1)\,(c\,p^{(2k+1)}-1)\,(c\,p^{(2k+3)}-1)\,(a\,p^{(2k+1)}-1)\,(a\,p^{(2k+3)}-1)\,d(k+2)=0$$

The following command computes two linearly independent q-hypergeometric solutions of the recurrence equation RE using the Petkovsek-Horn algorithm:

```
> TIME:=time():res:=qHypergeomSolveRE(RE,d(k),var=p,qsimplified=false,output=basis);
time()-TIME;
```

$$res := \left[2^k\,\mathrm{qpochhammer}\left(\frac{1}{p^n},p,k\right)\mathrm{qpochhammer}\left(-\frac{1}{p^n},p,k\right)\mathrm{qpochhammer}\left(-a\,c\,(p^n)^2,p^2,k\right)p^{(k(k+2))}\,e^{(-I\,k\,\pi)}\Big/\left(\mathrm{qpochhammer}(a\,p,p^2,k)\,\mathrm{qpochhammer}(c\,p,p^2,k)\,\mathrm{qpochhammer}(-p,p,k)\,\mathrm{qpochhammer}(-I\,p,p,k)\,\mathrm{qpochhammer}(p\,I,p,k)\,\mathrm{qpochhammer}(p,p,k)\right),\ 2^k\,\mathrm{qpochhammer}\left(a\,c\,(p^n)^2,p^2,k\right)\mathrm{qpochhammer}\left(\frac{-I}{p^n},p,k\right)\mathrm{qpochhammer}\left(\frac{I}{p^n},p,k\right)p^{(k(k+2))}\,e^{(-I\,k\,\pi)}\Big/\left(\mathrm{qpochhammer}(a\,p,p^2,k)\,\mathrm{qpochhammer}(c\,p,p^2,k)\,\mathrm{qpochhammer}(-p,p,k)\,\mathrm{qpochhammer}(-I\,p,p,k)\,\mathrm{qpochhammer}(p\,I,p,k)\,\mathrm{qpochhammer}(p,p,k)\right)\right]$$

$$83.959$$

The following term is our output in Eq. (62) and we show in the sequel that the two coefficients of Fn(x) agree:

```
> printedres1:=qpochhammer(q^(-(1/2)*n),q^(1/2),k)*qpochhammer(-a*c*q^((1/2)*n),q^(1/2),k)*(-2)^k*q^((1/4)*k*(k+2))/(qpochhammer(q^(1/2),q^(1/2),k)*qpochhammer(-q^(1/2),q^(1/2),k)*qpochhammer(a*q^(1/4),q^(1/2),k)*qpochhammer(c*q^(1/4),q^(1/2),k));
```

$$printedres1 := \mathrm{qpochhammer}\left(q^{\left(-\frac{n}{2}\right)},\sqrt{q},k\right)\mathrm{qpochhammer}\left(-a\,c\,q^{\left(\frac{n}{2}\right)},\sqrt{q},k\right)(-2)^k$$

$$q^{\left(\frac{k(k+2)}{4}\right)} \Big/ \left(\text{qpochhammer}(\sqrt{q}, \sqrt{q}, k)\, \text{qpochhammer}(-\sqrt{q}, \sqrt{q}, k)\, \text{qpochhammer}(a\, q^{(1/4)}, \sqrt{q}, k)\, \text{qpochhammer}(c\, q^{(1/4)}, \sqrt{q}, k)\right)$$

```
> rat1:=(simplify(subs(q=p^4,printedres1),power) assuming
  p>0)/res[1];
```

$$\text{rat1} := \text{qpochhammer}(p^{(-2\,n)}, p^2, k)\, \text{qpochhammer}(-a\, c\, p^{(2\,n)}, p^2, k)\, (-1)^k\, \text{qpochhammer}(-p, p, k)\, \text{qpochhammer}(-I\, p, p, k)\, \text{qpochhammer}(p\, I, p, k)\, \text{qpochhammer}(p, p, k) \Big/ \left(\text{qpochhammer}(p^2, p^2, k)\, \text{qpochhammer}(-p^2, p^2, k)\, \text{qpochhammer}\left(\frac{1}{p^n}, p, k\right) \text{qpochhammer}\left(-\frac{1}{p^n}, p, k\right) \text{qpochhammer}\left(-a\, c\, (p^n)^2, p^2, k\right) e^{(-I\, k\, \pi)}\right)$$

```
> qsimpcomb(subs(k=k+1,rat1)/rat1, QDE);
```

$$1$$

```
> qsimpcomb(subs(k=0,rat1), QDE);
```

$$1$$

The two above computations show that rat1(k+1)=rat1(k) and that rat1(0)=1, therefore rat1(k)=1. The second representation in Eq. (62) follows directly using the hypergeometric representation of Fk(x), see Eq. (52).

The following term is our output in Eq. (63) and we show in the sequel that the two coefficients of Fn(x) agree:

```
> printedres2:=qpochhammer(-q^(-(1/2)*n),q^(1/2),k)*qpochhammer(a*
  c*q^((1/2)*n),q^(1/2),k)*(-2)^k*q^((1/4)*k*(k+2))/(qpochhammer(q
  ^(1/2),q^(1/2),k)*qpochhammer(-q^(1/2),q^(1/2),k)*qpochhammer(a*
  q^(1/4),q^(1/2),k)*qpochhammer(c*q^(1/4),q^(1/2),k));
```

$$\text{printedres2} := \text{qpochhammer}\left(-q^{\left(-\frac{n}{2}\right)}, \sqrt{q}, k\right) \text{qpochhammer}\left(a\, c\, q^{\left(\frac{n}{2}\right)}, \sqrt{q}, k\right) (-2)^k\, q^{\left(\frac{k(k+2)}{4}\right)} \Big/ \left(\text{qpochhammer}(\sqrt{q}, \sqrt{q}, k)\, \text{qpochhammer}(-\sqrt{q}, \sqrt{q}, k)\, \text{qpochhammer}(a\, q^{(1/4)}, \sqrt{q}, k)\, \text{qpochhammer}(c\, q^{(1/4)}, \sqrt{q}, k)\right)$$

```
> rat2:=(simplify(subs(q=p^4, printedres2), power) assuming
  p>0)/res[2];
```

$$\text{rat2} := \text{qpochhammer}(-p^{(-2\,n)}, p^2, k)\, \text{qpochhammer}(a\, c\, p^{(2\,n)}, p^2, k)\, (-1)^k\, \text{qpochhammer}(-p, p, k)\, \text{qpochhammer}(-I\, p, p, k)\, \text{qpochhammer}(p\, I, p, k)\, \text{qpochhammer}(p, p, k) \Big/ \Big(\text{qpochhammer}(p^2, p^2, k)\, \text{qpochhammer}(-p^2, p^2, k)$$

$$\mathrm{qpochhammer}\left(a\,c\,(p^n)^2, p^2, k\right)\,\mathrm{qpochhammer}\left(\frac{-I}{p^n}, p, k\right)\,\mathrm{qpochhammer}\left(\frac{I}{p^n}, p, k\right) e^{(-I\,k\,\pi)}\Bigg)$$

```
> qsimpcomb(subs(k=k+1,rat2)/rat2, QDE);
```

$$1$$

```
> qsimpcomb(subs(k=0,rat2), QDE);
```

$$1$$

The two above computations show that rat2(k+1)=rat2(k) and that rat2(0)=1, therefore rat2(k)=1. The second representation in Eq. (62) follows directly using the hypergeometric representation of Fk(x), see Eq. (52).

```
>
```